

This is a high-level specification of *SCP* focusing on the nomination protocol.

Currently, as implemented, before voting for a txset hash, nodes wait to obtain its preimage. Delaying the point at which we wait for the pre-image would leave more room for disseminating the txset in parallel to nomination. However this has to be done carefully to maintain the main property of nomination: assuming that there is a nomination round with a good leader and during which the network is fast enough, at least a Tier-1 quorum must eventually enter *balloting*.

In the version specified in this document, we do not wait on the pre-image to vote for a txset hash, but we do wait for the pre-image before accepting it.

In the previous version of this document, we even accepted without a pre-image. There is a problem with this: it could create a situation in which not enough nodes can start *balloting* (*i.e.* not a full quorum) and the whole system is stuck.

The problem stems from the fact that, in the nomination protocol, nodes that confirm a candidate then stop voting for new values (otherwise nomination is not guaranteed to converge). So if a blocking set *B* confirms a candidate but somehow other nodes cannot get the pre-images they need to do so, more nomination rounds will not help because the members of *B* have stopped voting, which blocks the progress of any new candidate. Depending on how pre-images are disseminated, this can potentially be exploited by an attacker to halt the system.

So accepting without a pre-image is only workable if there is some way to guarantee that, once a Tier-1 blocking set has a pre-image, then everybody in Tier-1 eventually gets it.

Another problem is that we want it to be likely that a quorum starts *balloting* already in agreement and roughly at the same time. If we delay checking pre-images to the confirm stage, an attacker could first send the pre-image to a set *A* of nodes, which then enter *balloting* at time  $T_A$ , but not send the pre-image to another set *B* of nodes, which then enter *balloting* at time  $T_B \gg T_A$  because they need to get the pre-image from *A* before starting *balloting*. For example, if it takes 500ms for members of *B* to get the pre-image from members of *A*, then  $T_B = T_A + 500ms$ . This can cause the first ballot to end without a decision. Members of *B* could also start a new nomination round before  $T_B$  and then enter *balloting* not only late but also with a different value than members of *A*.

EXTENDS *Naturals*, *FiniteSets*

CONSTANTS

*V*, validators

*TxSet*, blocks

*Bot*, default value

*Quorum*(*\_*), *Quorum*(*v*) is the set of quorums of validator *v*

*Blocking*(*\_*), *Blocking*(*v*) is the set of blocking sets of validator *v*

*Combine*(*\_*), the functions that combines candidates to produce a txset for *balloting*

*H*, domain of hashes

*Hash*(*\_*) hash function

--algorithm *SCP*{

variables global variables (*e.g.* representing messages or cross-component variables like *ballotingTxSet*):

*ballotingTxSet* = [*v* ∈ *V* ↦ *Bot*]; for each validator, the nominated txset for *balloting*

*decision* = [*v* ∈ *V* ↦ *Bot*]; for each validator, the *balloting* decision

*voted* = [*v* ∈ *V* ↦ {}]; X in the whitepaper (nomination Section)

```

    accepted =  $[v \in V \mapsto \{\}];$   Y in the whitepaper (nomination Section)
process ( nomination  $\in V$  )
    variables    local variables:
        round = 0;  nothing happens in round 0; the protocol start at round 1
        candidates =  $\{\};$   Z in the whitepaper (nomination Section)
        preImage =  $[h \in H \mapsto Bot];$   the pre-images the validator knows about
        leader = Bot;  leader for the current round
    {
ln1:  while ( TRUE )
    either {  timeout and go to next round (this also starts round 1)
        round := round + 1;
        with ( l  $\in V$  ) {  pick a leader
            leader := l;
            if ( l = self )  if the leader is the current node, pick a txset and vote for it
            with ( txs  $\in TxSet$  ) {
                preImage[Hash(txs)] := txs;
                voted[self] := voted[self]  $\cup \{Hash(txs)\}$ 
            }
        }
    }
    or if ( candidates =  $\{\}$  ) {  vote for what the leader voted for, unless we have a candidate already
        when leader  $\neq Bot$ ;
        with ( hs = voted[leader] ) {
            await hs  $\neq \{\}$ ;  wait to hear from the leader
            voted[self] := voted[self]  $\cup hs$   vote for what the leader has voted for
            in the whitepaper version, we would only vote for the hashes for which we have a pre-image:
            with (hsWithPreimage =  $\{h \in hs : preImage[h] \neq Bot\}$ )
            voted[self] := voted[self]  $\cup hsWithPreimage$ 
        }
    }
    or with ( Q  $\in Quorum(self)$ , h  $\in H$  ) {  accept when voted or accepted by a quorum and we have the pre-im
        when preImage[h]  $\neq Bot$ ;  we must have received the block
        when  $\forall w \in Q : h \in voted[w] \vee h \in accepted[w];$   a quorum has voted or accepted h:
        accepted[self] := accepted[self]  $\cup \{h\};$   accept h
    }
    or with ( Bl  $\in Blocking(self)$ , h  $\in H$  ) {  accept when accepted by a blocking set and we have the pre-image
        when preImage[h]  $\neq Bot$ ;  we must have received the block
        when  $\forall w \in Bl : h \in accepted[w];$ 
        accepted[self] := accepted[self]  $\cup \{h\};$   accept h
    }
    or with ( txs  $\in TxSet$  ) {  receive a txset
        preImage[Hash(txs)] := txs;
    }
    or with ( Q  $\in Quorum(self)$ , h  $\in H$  ) {  confirm b as candidate
        when preImage[h]  $\neq Bot$ ;  we must have received the block

```

```

    when  $\forall w \in Q : h \in \text{accepted}[w]$ ; a quorum has accepted  $h$ :
      candidates := candidates  $\cup \{\text{preImage}[h]\}$ ; add  $h$  to the confirmed candidates
      update the block used in balloting:
      ballotingTxSet[self] := Combine(candidates); this starts the balloting protocol (see below)
    }
  }
  as a first approximation, balloting just decides on one of the balloting blocks:
  note we cannot reuse the process ID identifiers used in nomination, so we add the “balloting” tag
  process ( balloting  $\in \{\langle v, \text{“balloting”}\rangle : v \in V\}$  ) {
lb1:   await ballotingTxSet[self[1]]  $\neq \text{Bot}$ ; wait for a confirmed candidate from nomination
lb2:   with (  $b \in \{\text{ballotingTxSet}[v] : v \in V\} \setminus \{\text{Bot}\}$  ) {
      when  $\forall w \in V : \text{decision}[w] \neq \text{Bot} \Rightarrow b = \text{decision}[w]$ ;
      decision[self[1]] :=  $b$ ;
    }
  }
}

```

BEGIN TRANSLATION ( $\text{chksum}(\text{pcal}) = \text{“b24885fb”} \wedge \text{chksum}(\text{tla}) = \text{“95d64d8b”}$ )  
 VARIABLES *ballotingTxSet*, *decision*, *voted*, *accepted*, *pc*, *round*, *candidates*,  
*preImage*, *leader*

$\text{vars} \triangleq \langle \text{ballotingTxSet}, \text{decision}, \text{voted}, \text{accepted}, \text{pc}, \text{round}, \text{candidates}, \text{preImage}, \text{leader} \rangle$

$\text{ProcSet} \triangleq (V) \cup (\{\langle v, \text{“balloting”}\rangle : v \in V\})$

$\text{Init} \triangleq$  Global variables  
 $\wedge \text{ballotingTxSet} = [v \in V \mapsto \text{Bot}]$   
 $\wedge \text{decision} = [v \in V \mapsto \text{Bot}]$   
 $\wedge \text{voted} = [v \in V \mapsto \{\}]$   
 $\wedge \text{accepted} = [v \in V \mapsto \{\}]$   
Process nomination  
 $\wedge \text{round} = [\text{self} \in V \mapsto 0]$   
 $\wedge \text{candidates} = [\text{self} \in V \mapsto \{\}]$   
 $\wedge \text{preImage} = [\text{self} \in V \mapsto [h \in H \mapsto \text{Bot}]]$   
 $\wedge \text{leader} = [\text{self} \in V \mapsto \text{Bot}]$   
 $\wedge \text{pc} = [\text{self} \in \text{ProcSet} \mapsto \text{CASE } \text{self} \in V \rightarrow \text{“ln1”}$   
 $\quad \square \quad \text{self} \in \{\langle v, \text{“balloting”}\rangle : v \in V\} \rightarrow \text{“lb1”}]$

$\text{ln1}(\text{self}) \triangleq$   $\wedge \text{pc}[\text{self}] = \text{“ln1”}$   
 $\wedge \vee \wedge \text{round}' = [\text{round EXCEPT } ![\text{self}] = \text{round}[\text{self}] + 1]$   
 $\wedge \exists l \in V :$   
 $\quad \wedge \text{leader}' = [\text{leader EXCEPT } ![\text{self}] = l]$   
 $\quad \wedge \text{IF } l = \text{self}$   
 $\quad \text{THEN } \wedge \exists \text{txs} \in \text{TxSet} :$   
 $\quad \quad \wedge \text{preImage}' = [\text{preImage EXCEPT } ![\text{self}][\text{Hash}(\text{txs})] = \text{txs}]$   
 $\quad \quad \wedge \text{voted}' = [\text{voted EXCEPT } ![\text{self}] = \text{voted}[\text{self}] \cup \{\text{Hash}(\text{txs})\}]$

```

ELSE  $\wedge$  TRUE
     $\wedge$  UNCHANGED  $\langle voted, preImage \rangle$ 
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, accepted, candidates \rangle$ 
 $\vee \wedge$  IF  $candidates[self] = \{\}$ 
    THEN  $\wedge leader[self] \neq Bot$ 
         $\wedge$  LET  $hs \triangleq voted[leader[self]]$  IN
             $\wedge hs \neq \{\}$ 
             $\wedge voted' = [voted \text{ EXCEPT } ![self] = voted[self] \cup hs]$ 
    ELSE  $\wedge$  TRUE
         $\wedge voted' = voted$ 
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, accepted, round, candidates, preImage, leader \rangle$ 
 $\vee \wedge \exists Q \in Quorum(self) :$ 
     $\exists h \in H :$ 
         $\wedge preImage[self][h] \neq Bot$ 
         $\wedge \forall w \in Q : h \in voted[w] \vee h \in accepted[w]$ 
         $\wedge accepted' = [accepted \text{ EXCEPT } ![self] = accepted[self] \cup \{h\}]$ 
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, voted, round, candidates, preImage, leader \rangle$ 
 $\vee \wedge \exists Bl \in Blocking(self) :$ 
     $\exists h \in H :$ 
         $\wedge preImage[self][h] \neq Bot$ 
         $\wedge \forall w \in Bl : h \in accepted[w]$ 
         $\wedge accepted' = [accepted \text{ EXCEPT } ![self] = accepted[self] \cup \{h\}]$ 
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, voted, round, candidates, preImage, leader \rangle$ 
 $\vee \wedge \exists txs \in TxSet :$ 
     $preImage' = [preImage \text{ EXCEPT } ![self][Hash(txs)] = txs]$ 
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, voted, accepted, round, candidates, leader \rangle$ 
 $\vee \wedge \exists Q \in Quorum(self) :$ 
     $\exists h \in H :$ 
         $\wedge preImage[self][h] \neq Bot$ 
         $\wedge \forall w \in Q : h \in accepted[w]$ 
         $\wedge candidates' = [candidates \text{ EXCEPT } ![self] = candidates[self] \cup \{preImage[self][h]\}]$ 
         $\wedge ballotingTxSet' = [ballotingTxSet \text{ EXCEPT } ![self] = Combine(candidates'[self])]$ 
 $\wedge$  UNCHANGED  $\langle voted, accepted, round, preImage, leader \rangle$ 
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = "ln1"]$ 
 $\wedge$  UNCHANGED  $decision$ 

```

$nomination(self) \triangleq ln1(self)$

$lb1(self) \triangleq$   $\wedge pc[self] = "lb1"$   
 $\wedge ballotingTxSet[self[1]] \neq Bot$   
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = "lb2"]$   
 $\wedge$  UNCHANGED  $\langle ballotingTxSet, decision, voted, accepted, round,$   
 $candidates, preImage, leader \rangle$

$lb2(self) \triangleq$   $\wedge pc[self] = "lb2"$   
 $\wedge \exists b \in \{ballotingTxSet[v] : v \in V\} \setminus \{Bot\} :$

$$\begin{aligned}
& \wedge \forall w \in V : decision[w] \neq Bot \Rightarrow b = decision[w] \\
& \wedge decision' = [decision \text{ EXCEPT } ![self[1]] = b] \\
& \wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"Done"}] \\
& \wedge \text{UNCHANGED } \langle ballotingTxSet, voted, accepted, round, \\
& \quad candidates, preImage, leader \rangle
\end{aligned}$$

$$balloting(self) \triangleq lb1(self) \vee lb2(self)$$

$$\begin{aligned}
Next & \triangleq (\exists self \in V : nomination(self)) \\
& \vee (\exists self \in \{v, \text{"balloting"}\} : v \in V : balloting(self))
\end{aligned}$$

$$Spec \triangleq Init \wedge \Box [Next]_{vars}$$

END TRANSLATION

The type-safety invariant:

$$\begin{aligned}
TypeOkay & \triangleq \\
& \wedge ballotingTxSet \in [V \rightarrow TxSet \cup \{Bot\}] \\
& \wedge decision \in [V \rightarrow TxSet \cup \{Bot\}] \\
& \wedge voted \in [V \rightarrow \text{SUBSET } H] \\
& \wedge accepted \in [V \rightarrow \text{SUBSET } H] \\
& \wedge round \in [V \rightarrow Nat] \\
& \wedge candidates \in [V \rightarrow \text{SUBSET } TxSet] \\
& \wedge preImage \in [V \rightarrow [H \rightarrow TxSet \cup \{Bot\}]] \\
& \wedge leader \in [V \rightarrow V \cup \{Bot\}]
\end{aligned}$$

Next we specify a liveness property that we can easily check with the *TLC* model-checker.

This property is that, if a validator  $v$  enters *balloting*, then eventually all validators enter *balloting*. This will hold in simple configurations where the whole network is top tier.

For the property to hold, we also need to add fairness assumptions (*e.g.* if a node can vote for a value, it will eventually do so). Unfortunately the TLA+ code generated from the *PlusCal* specification is in a form that makes stating fairness assumptions hard. It seems that we would need to rewrite this in pure TLA+ to tackle liveness.

$$\begin{aligned}
NominationLiveness & \triangleq \\
& \forall v, w \in V : \Box(ballotingTxSet[v] \neq Bot \Rightarrow \Diamond(ballotingTxSet[w] \neq Bot))
\end{aligned}$$

Definition for model-checking:

Concrete hashing for the model-checker:

$$TestH \triangleq 1 \dots Cardinality(TxSet)$$

$$TestHash(b) \triangleq$$

$$\begin{aligned}
& \text{LET } f \triangleq \text{CHOOSE } f \in [TxSet \rightarrow H] : \forall txs1, txs2 \in TxSet : txs1 \neq txs2 \Rightarrow f[txs1] \neq f[txs2] \\
& \text{IN } f[b]
\end{aligned}$$

Debugging canaries:

$$Canary1 \triangleq \forall v \in V : decision[v] = Bot$$

$Canary2 \triangleq \forall v \in V : Cardinality(candidates[v]) \leq 1$

$Canary3 \triangleq \forall v \in V : ballotingTxSet[v] = Bot$

$TestQuorums \triangleq \{Q \in SUBSET\ V : 2 * Cardinality(Q) > Cardinality(V)\}$

$TestBlocking \triangleq \{Bl \in SUBSET\ V : Cardinality(Bl) > 1\}$

---

\ \* Modification History

\ \* Last modified *Thu Mar 30 20:16:04 PDT 2023* by *nano*

\ \* Created *Thu Mar 30 20:15:37 PDT 2023* by *nano*